
Error handling is an integral part of any application because all applications have natural and unnatural error conditions triggered as the users exercise the software. Visual Fox Pro has several techniques available to address these errors. Which is better to use? Should you implement an all encompassing, monolithic global error handler? Is the encapsulated Error method the route to go? Should you invest the time to modernize the architecture using the structured error handling introduced in VFP 8? Is a blend of the various options the best practice you should adopt? How are you going to be informed when the applications you support don’t work as advertised?
Introduction

Error handling is a fundamental aspect of any software development project. It does not matter if the application runs on one desktop or millions of desktops. It does not matter if it is a Web application, or a Web Service. It does not matter if it is a custom application or something written to serve a vertical market. Plain and simple: errors happen. They happen naturally (printers not available, disconnected network connections), and sometimes they happen because you and I make mistakes in the code we craft.
This session will provide several best practice ideas to address the following issues:

1. Addressing the errors in your applications.

Demonstrating all the error handling techniques available in Visual Fox Pro and discuss when you might use one over the other or integrate some together.

Determining the information you can trap so you know what went wrong, and where the user was when the trouble occurred.

Capturing the details and key information surrounding the error.

Communicating errors back to the development team so the problem can be fixed if necessary.

Tracking errors once they are reported.

Errors happen

Application errors are a fact of life and a developer who claims their applications don’t have errors is not writing code. Errors happen naturally and predictable. They also happen unnaturally and are not always predictable. You cannot avoid the errors happening naturally, but you can program defensively to help control the amount of errors happening unnaturally.
Programmer errors

All developers make boneheaded mistakes when code is written. It is only human to err. You create invalid VFP syntax, you pass two many parameters to routines, you pass parameters in the incorrect order, or you try to replace character data into a numeric column. Bad code happens all the time and a good error handler will capture the errors if the code is completely tested. Unfortunately, we do not always test every single test case. More often than not, we do not build all the test cases needed to fully exercise the code during testing, and our users find the problems.
Predictable errors
Predictable errors are the type of errors you expect to happen and handle appropriately when the application triggers them. You know errors can happen when the application encounters a corrupt index, a corrupt table, a missing file, or a situation where a network resource disappears. You know errors are triggered when files are in use, locked, a field or row level rule is violated, or you add a record and attempt to duplicate the primary key. We know printers are not always available, and automation objects get unregistered. Each of these errors has to be addressed and can be addressed in the application using the various error handling capabilities VFP gives us.
Unpredictable errors
The unpredictable errors are the errors you don’t know about in advance, or more importantly don’t address in your error handler directly, and you log with the details needed to determine what went wrong. This is the OTHERWISE part of the code in the long CASE statement of errors handled and directly addressed. 
Real world stories

Implementing error handling for the first time in an application can be painful. I was hired by another developer to write an error handler. I had one I was using in my FoxPro DOS applications and provided it to him. Thirty minutes after I sent it to him he called me up and complained his application was not working. It was displaying a red box (I used a red window to display the error to get the users attention. His complaint was centered on the fact that it was capturing too many errors. He fixed several but could not get the app to even start up in thirty minutes. He abandoned the error handler in favor of running code ignoring all errors.
I was part of a team working on an application migration from a VFP database with DBFs to SQL Server on the backend. At one point we had trouble with one of the forms not working. Upon inspection we found bad code, which should have triggered an error, but did not. We could not figure out why the error handler had not kicked in and informed us of the problem. After much searching we found the form manager had code in it to turn off all error handling. This means the entire application was running with no error handler once a form was started. I guess this is one way to impress your clients when it comes to bug free code.

Another example I had in my career was a moment in time where I learned a lot about the quality of developers and architects on the first n-tier application I worked on. I was brought in on this project to help with performance of a reporting module. The application was developed with a Visual Basic user interface, a C++ middle tier, and Sybase as the backend database. Part of my job was to cycle in on the tech support team for a week at a time. I was working late one night when a call came in from a user. The screen shot of the error said something to the effect of Error #, another user has changed the same data, app is shutting down. The only developer available was one of the VB guys. I inquired about the error. He told me it happens all the time and is “normal.” The middle tier guys return the error when two or more people are editing the same data so one user has to lose their changes. I was naturally flabbergasted by the design and the willingness of the developers involved to just kill the app.
User response

It is my experience the users will hide an error notification as long as possible. Sure some users will dial up the technical support line within seconds of seeing the error dialog presented to them just so they can take the opportunity to see if they can make the help desk person breakdown in tears. I am no longer surprised when I review the contents of an error log file to find numerous entries not reported back to me or my tech support professionals.

There are a number of reasons for this behavior, but it is mostly the thought of someone catching them doing something wrong. They do not want to be in trouble for breaking the expensive software their company purchased to make their job easier. While I am tempted to explain to the CIO the reason my application came to a crashing halt was the fault of the user, 100 percent of the time it is either something I wrote in my code, or something I did not write in my code causing the problem.
So the trick is to record the problem, collect as many details about the problem, and provide a mechanism to transfer this log back to the people who can make the problem go away. Be resigned to the possibility users often will not tell you the problem is happening unless it is a showstopper and makes it so they cannot do their job.
Application response
Not all errors are equally serious. The application response to an error condition is one way you can distinguish yourself and your products from your competition. The key in all error conditions is to handle the error gracefully. Crashing your application hard and taking out tables or corrupting data is something no one enjoys and will only make the technical support mop up more costly in respect to loss time for you and your customer.
One short disclaimer before I discuss the different application responses to errors: the errors discussed in each section are a sampling of errors in each classification. It is not my intent to discuss every possible error, nor provide all inclusive coverage of the classifications. The VFP 9 Help file documents over 870 errors and it is not practical for any VFP developer to address all these errors in their applications.
Address without notification
There are certain errors which happen in the natural course of working with an application where no user notification has to take place, and no response is needed. You might log the error for diagnostics. These errors are not severe and are more along the level of a serious warning to you. It could be something of an advanced warning of bigger problems to come.
· Record is locked by another user error (109, 1502)
· Primary key already exists (1883)
Address with notification

Visual FoxPro triggers errors when validation fails. Our applications need to address this. The error hander is not the place to address the various issues, but it does have to ignore the errors and the application will need to deal with them. Or the application needs to address the problems before committing the data. 
· Field validation rule is violated (1582)
· Record validation rule is violated (1583)
Try again

Some errors are a timing thing and the next time the code is run the error condition does not exist. When this is the case we can present the users with a message telling them what is wrong and have them try again. One of the problems with this is the user can tied into a resource contention issue with other users. If this is not resolved the user needs the option to cancel the operation.
· Printer errors (124, 125, 1956, 1643, 1644)
· Record is not locked (130)
· File access denied, File is in use (1705, 108)
· File is not locked (1503)
Avoid

Avoiding errors sounds like one of the “duh” things, but errors can happen based on conditions outside of the application. Errors can happen because the users do not have other applications loaded (like a spreadsheet program your application automates), drive not available, or the Internet connection gets disconnected and your application needs to connect to a Web service. The rest of the application might run fine, but these errors need to be captured, the user informed of the situation and possibly the program advises them of some corrective action to take. Depending on the type of user and the type of application you might have to tell the user other parts of the application are working fine:
· OLE errors (1420, 1421, 1422, 1423, 1424, 1426, 1427, 1428, 1429, 1431, 1434, 1436, 1440, 1441, 1442, 1443, 1444, 1508)
· EventHandler errors (1445)

Abandon

You can code defensively all day and still not have time to address every single error VFP can throw as your application runs. There is a point where the return on the investment of coding time is not worth the effort because the error happens so infrequently. Then there are the errors that happen and there really is nothing the user can do to take corrective action. When this happens you cut your losses and record the details in your error handler. Some of the more common errors captured in this category are:
· Error instantiating the object or the form class doesn't exist (1736, 1733)
· Data type mismatching, syntax error, and code limitations (9, 10, 11, 12, 16, 31, 36, 37, 42, 46, 107, 1211)
· Table, index, alias, and corruption issues (13, 14, 19, 20, 34, 41, 112, 114, )
· Exceeding internal VFP limitation (18, 21, 23, 39, 96, 103, 1201, 1202, 1212)
· Too many files open (6)
· Environment limitations and memory problems (43, 56, 1149, 1150, 1151, )
· Feature not available (1001)

· Internal consistency error (1000)

· Divide by zero (1307)
Error handling fundamentals

Before I jump into the types of error handling available to you in Visual FoxPro I want to address the fundamentals of a good error handler. Every error handler needs to capture errors and address them as detailed in the previous section. The complete strategy, no matter how you implement it in code needs to provide services to notify the user of the problem, capture details about the error and log it to a file, and if serious enough to shut off the application – clean up the program and bring it down gracefully.
Determining the cause of the error

There are several functions in Visual FoxPro you need to understand when working with an error handler. The key ones are ERROR(), MESSAGE(), LINENO(), PROGRAM(), SYS(16), SYS(2018), SYS(2410), and AERROR(). You will use each of these in the different types of error handling schemes available in Visual FoxPro. More details on when you might use these commands are discussed later in this whitepaper.
Notification

Some developers take the minimalist approach and tell the user an error occurred and the program is shutting down. I have seen other developers push the equivalent of the blue screen of death and a meaningless hex dump to the screen. I believe there is a happy medium.

If the problem is serious and the application is going to be shut down the user needs to know. They need to know why the application is about to apply some pain to their day. They also need to know who to call and how to report the problem. If appropriate, they also need to now how they can fix the problem.
I have read some developer opinions on how the user should never see raw error numbers and the details you need to help track down the problem. Each situation will dictate what you display and don’t display, but in general I think it is important to put as much information on the screen so the user can immediately call for help and get the ball moving on a resolution.
Be assertive

Notifying the user in the production run time application is an obvious must have, but what about notifying developers of errors as they do unit testing, integration testing, and system testing?
The normal error handling is available in the Visual FoxPro IDE, but as I have discussed earlier in this whitepaper, not all errors are reported to the user though a user interface component. This means errors which are automatically handled for end users go through a notification process for developers. You can see the errors and address them as necessary during testing.
The easiest way to do this is through the ASSERT command. The best thing about using an assertion is they can be turned on and off by the developer before running the test so the messages (Figure 1) are not always in the way. They have no effect on the run time version of your application so the users will never see them in the production environment (unless your users run your applications in Visual FoxPro). Here is a sample of what you can do:
LPARAMETERS tnError, tcMethod, tnLine, tcCallingObject

AERROR(laError)

lcMessage = "Error: (" + ALLTRIM(STR(tnError)) + ") '" +  laError[2] + ;

            "' happened in " + tcCallingObject + "." + ;

            tcMethod + "()" + ;

            " on line " + ALLTRIM(STR(tnLine))
ASSERT .F. MESSAGE (lcMessage)

Figure 1. The ASSERT command produces a dialog with handy options for developers when testing an application in Visual FoxPro.

Each time an error is triggered you get the option to jump into the debugger, cancel the program, ignore the error, or turn off future assertions with the Ignore All. 
Error Log

The error log is your best friend when it comes to solving problems. It is invaluable. I discuss specifics on what you can capture in an error log later in the section titled “Logging errors.” The error log should contain anything you find helpful to tracking down the cause of the error. An error log can be a simple text file or set of text files, a DBF, or even a table in the backend SQL database.
The simple text file approach has advantages in the situations where VFP is having trouble writing to a table. I cannot recall ever running into the situation where VFP could not dump to a text file, but the possibility definitely exists. The text files can be zipped up and sent to the developers or the developers can connect to the customer’s site and read the log with something as simple as NotePad. Integrating the error log review feature into the application is not as clean as a table, but with a function like FILETOSTR() it is not difficult.
The table approach has some obvious benefits when it comes to building the user interface to present the log to the user or for tech support. The other key thing I use it for is to run summary and analysis queries on the table. How many times did the application crash for user X? How many times did the users run into a printer error? Collecting this information can help you make changes to the application so it is more stable, or appear more stable because less errors are captured to the log.

The approach I have taken is a mixture of text files and a table. I first dump out the information I am collecting in the error handler to a text file and use FILETOSTR() to pull the details into a memo field in the table.
One best practice when it comes to error logs I find important is ensuring the application creates an empty error log if one does not exist when the program is started. Error logs can get big depending on the amount of detail you record when the errors are handled. Once I have logged the problems reported to me I like to delete the error log to free up space, and to start with a fresh set of logged issues. If the application does not have an error log and the error handler depends on it being available you have a problem. For this reason I have the error log created if it is missing.
Alternatively you can have the error handler create the error log when the first error occurs. The problem with this approach is the possibility that VFP is not capable of creating a table (an error condition) and therefore cannot log the error, which triggered the error handler.

Clean shut down

The state of the application can vary for each error condition. The error handler has to be extremely smart when it comes to “undoing” the application state and shutting down. Things you should consider:
· Shut off the ON ERROR
· Rollback any open transactions

· Revert any dirty buffers

· Close all open forms and windows

· Release all global objects (application object, forms manager, menu manager, reports manager, etc.)

· Restore the system menu

· Shut off the ON SHUTDOWN
Types of error handling in VFP

Entire whitepapers have been written on each of the different types of error handling (and are documented in the Resources section towards the end of this whitepaper). It is not the purpose of this section to provide you with all the ins and outs of the different types of error handling, but to show you how you might use them in your applications.
Default error handling

Visual FoxPro applications which do not have any error handling scheme get the default error handler built into the product. You can achieve this behavior by never coding a TRY…CATCH, adding any code to any object’s Error event, and by not specifying an ON ERROR handler.
The default error handler is a message box (see Figure 2) with the error text and a few buttons. In the Interactive Development Environment you get 
1. Cancel – quits the program and returns to the IDE if you are running in the IDE

Suspend – stops the program on the offending line of code, you enter the debugger
Ignore – passes by the error and continues on the next line of code

Help – opens up the Help file if Help is on and shows the topic for the error

The executable does not get the Suspend option and the Help will not show you anything unless you have all the error messages migrated from the VFP Help file into your application Help file.

Figure 2. The default error handler message in the VFP IDE is on the left and the same message in an EXE is on the right.

You can shut off the default error handler to avoid all errors being handled by executing the following code:

ON ERROR * 

I do not recommend relying on the default error handler for many reasons. It only provides you with two choice (Cancel, Ignore), and does nothing to describe where the error occurred, what the offending code is, and does nothing to record the state of the program where the problem reared its ugly head. It also provides your users the raw error, which might not make any sense to them.
Global 
The ON ERROR command is the global error handler enabler and has been in FoxPro since day one. This is also referred to as a procedural error handler and is the most common of the error handler approaches. The reason for this is the longevity of the approach and the amount of code developers have written over the years.
ON ERROR ;

     DO ErrHand.prg WITH ;

          ERROR(), MESSAGE(), MESSAGE(1),;

          PROGRAM(), LINENO(), SYS(2018)
PARAMETER tnError, tcMessage, tcMessage1, ;

          tcProg, tnLineNo, tcMsgParameter
The advantages include a centralized set of code to manage all errors, is the error handler of last resort if the other two styles are not used in particular situations, and there are battle tested code bases developed in production applications. 
The disadvantages include the monolithic design does not handle errors well for customized situations, the error handler code is removed from the code triggering the error, often has to be turned off to deal with specific types of problems, and in older versions of VFP (before VFP 9) could not capture errors in reports. 
It is important to further discuss the fact the ON ERROR code runs removed from the code triggering the error. This happens because the global error handler is called outside of VFP’s event handler based on the ‘ON’ event style introduced in FoxPro. If the error happens in an object you cannot use references to the suspect object because the error handler code is running outside of the object. ON ERROR code typically runs in the default datasession. Errors occurring in a private data session will complicate how well the global error handler can deal with the errors.
One best practice with ON ERROR is to pass it as many parameters as you need to understand the error at the point it happens in the code. In the example code above I show the six most common parameters passed in to the error handler. The reason I like to pass these parameters in is it saves the state of the first error in question. If another error occurs in the error handler I still have the original state of the original error.
You need to make sure to shut off error handling inside the error handler. Otherwise you open up a recursive situation and this does not have a pretty conclusion. In fact, your error handling has to be robust and one of the most tested pieces of code in the framework or your application.

Another best practice is to include the AERROR function at the beginning of the error handler code to build the array with the details about the error. A lot of the details like the error number and the error message are duplicates of the parameters I pass in, but the AERROR function provides information about OLE Automation and data validation (like trigger failures) you are not able to capture with parameters passed to the ON ERROR routine. The combination of the two techniques will provide a clear picture of what the error is and what might be the root cause.
It is important to note the global error handler does not have to be in program code (PRG), but often you see it this way because it is the way we did things prior to Visual FoxPro. You can instantiate a class and set up the ON ERROR to call a method of the class, and pass in the expected parameters. Using a class has a huge advantage with the ability to be subclassed for different situations. This is common in frameworks where the developers might subclass the original framework error handler and enhance it to record new details, or additionally handle errors not considered by the framework vendor.
Naturally the global error handler needs to address as many of the errors as you feel can be handled. This is going to be a giant CASE statement with the error numbers and how you are going to deal with the errors. The second key is going to be the OTHERWISE clause where you collect the details about the environment and log the error in the error log file.
The global error handler is time tested and obviously the most used approach in error handling in Visual FoxPro applications. Many developers I have worked with in the past have only implemented this type of error handling in their applications. Solely counting on the global error handling in your application might not provide you with the biggest benefits as you will see when I discuss the other types of error handling.
Object-level
The Error event was introduced in Visual FoxPro 3.0 and is an event you can find on all base classes except the Empty. This is referred to as class or object-level error handling. The error handling is encapsulated in the object.
The concept of encapsulation should make it obvious your code in the Error event will handle errors for the object. Predictable errors get handled in the method code, and others are passed along to another object or the global error handler, or they can be ignored completely (not typically recommended).
I believe the easiest way to show the power of the object-level error handling is to discuss a real life implementation. You might have a class you developed to work with Microsoft Word Automation. This class establishes an object reference to Word, opens a Word document, adds some text to bookmarks, prints out the document, and closes down Word.
The error handler for this object needs to address several potential problems (see Listing 1). First of all Word might not be loaded on this machine and the CREATEOBJECT function will trigger an error. The Word document might not exist on the drive. The bookmarks might not be in the document. The printer selected in Word might not be available on the machine. So I have four errors identified for this class. If another error is trigger like a syntax error I want the global error handler to deal with it.

Included in the Developer Downloads for this whitepaper, available from the conference download site, is a class called cusWordAutomation in the ErrorEventExample.VCX.

Listing 1. This code is an example of Error event method for error handling.

LPARAMETERS tnError, tcMethod, tnLine

LOCAL lnRows, lnColumns, lnIndex

LOCAL ARRAY laError[1,7]

AERROR(laError)

lnRows    = ALEN(this.aErrorLog, 1)

lnColumns = ALEN(this.aErrorLog, 2)

IF VARTYPE(this.aErrorLog[1]) = "L" AND lnRows = 1

   * First error, do nothingto adjust array size

ELSE

   lnRows = lnRows + 1

   DIMENSION this.aErrorLog[lnRows, lnColumns]

ENDIF

* Add AERROR to collection of errors

FOR lnIndex = 1 TO ALEN(laError, 2)

   this.aErrorLog[lnRows, lnIndex] = laError[lnIndex]

ENDFOR

* Add on the method and the line number to the error log

this.aErrorLog[lnRows, lnIndex + 0] = tcMethod

this.aErrorLog[lnRows, lnIndex + 1] = tnLine

DO CASE

   * Deal with possibility Word is not available

   CASE tnError = 1733

      * Do nothing, reference to Word will be NULL and code handles this

      this.aErrorLog[lnRows, lnIndex + 2] = "Addressed"

   * Something wrong with the automation, 

   * In this case bad document name, bad printer name

   CASE tnError = 1429

      * Code addresses these problems

      this.aErrorLog[lnRows, lnIndex + 2] = "Addressed"

   * Member is not available (like the bookmark collection)

   CASE tnError = 1943

      * ... we need to take control of problems

      this.aErrorLog[lnRows, lnIndex + 2] = "Addressed"

   OTHERWISE

      * Superclass can deal with the error

      * - Could be generic code for more errors

      * - Could delegate to global error handler

      * - Could delegate to the container object via Parent reference

      this.aErrorLog[lnRows, lnIndex + 2] = "Delegated"

      DODEFAULT(tnError, tcMethod, tnLine)

ENDCASE

RETURN

You can see in the OTHERWISE I have elected to delegate the errors I am not handling to the superclass. I could have elected to pass it along to the global error handler or delegate it to the container class if one existed. I will discuss this more in the “Error Handling Strategy” section later in this whitepaper.

You can add error handling code to all your base classes to address any and all errors triggered in the classes. This does not mean you can skip a global error handler because not all code is run from classes. There is a good chance you still have procedural code in your applications.
One gotcha I have been burned on in the past is the priorities of the error handlers. I discuss this shortly in detail. The gotcha I am referring to is procedural code called from an object method that triggers an error will have the Error event code be first to handle the error.
Structured
The TRY…CATCH…FINALLY was introduced in Visual FoxPro 8. This is referred to as structured error handling. What this means is the code inside of a designated block of code (the TRY) runs and if there is an error triggered, another block of code in the structure runs so you can attempt to address the error. The new TRY…CATCH…FINALLY gives us localized error handing in any code.
The TRY block represents the code VFP is going to attempt to run. If no error is triggered the optional code in the FINALLY block of the structure is run, and the program continues with the code following the ENDTRY (the designator for the end of the TRY). If an error does occur, the CATCH statements are evaluated similarly to CASE statements, sequentially until one is evaluated true.
There is a small catch when it comes to structured error handling. You need to handle all errors in the CATCH clauses. If you don’t you get the dreaded “Unhandled Structured Exception” error (2059). The good news is you can handle these pesky errors with the global error handler (ON ERROR) or the Error event code inside an object. You can avoid unhandled structured exceptions with a CATCH-all using code like this: 

CATCH WHEN .T. 

Another way to trigger an unhandled exception is to have an error in your CATCH code. The key is to handle all errors because the original error isn’t handled if you trigger an unhandled structured exception. The code in Listing 2 shows the structured error handling with a number of errors in the TRY block. I implemented the global error handler to deal with the unhandled structured exceptions (divide by zero in TRY and undeclared array laArrayNotDefined in last CATCH). The program executes the error code in the TRY, works to the appropriate CATCH and displays the problem, then the FINALLY messagebox is displayed. If there is an unhanded structured exception the global error handler kicks in right after the ENDTRY.

Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called ShowTryCatchExample.PRG to demonstrate how the structured error handling works in Visual FoxPro.

Listing 2. This code demonstrates structured error handling in Visual FoxPro.

#DEFINE ccCRLF   CHR(13) + CHR(10)

LOCAL loException AS Exception, ;

      nThisParameterDoesNotExist

ON ERROR DO ErrHand WITH ;

          ERROR(), MESSAGE(), MESSAGE(1),;

          PROGRAM(), LINENO(), SYS(2018)

TRY

   nThisParameterDoesNotExist = 0

   *< CREATE TABLE Customer FREE LONGNAME ;

      (iCustPK   i, ;

       cCustName c(200))

   *< USE DoesNotExist EXCLUSIVE IN 0 

   ALEN(laError, 1, nThisParameterDoesNotExist)

   x = 100 / 0    && Unhandled divide by zero 

CATCH TO loException WHEN loException.ErrorNo = 10

   * Handle syntax errors

   WAIT WINDOW loException.Message + CHR(13) + loException.LineContents

CATCH TO loException WHEN loException.ErrorNo = 1

   * Handle file does not exist errors

   MESSAGEBOX(loException.Message + CHR(13) + loException.LineContents, ;

              0 + 48, ;

              "Structured Error Handling Example")

CATCH TO loException WHEN loException.ErrorNo = 1230

   * Handle Too many arguements

   MESSAGEBOX(loException.Message + CHR(13) + loException.LineContents, ;

              0 + 48, ;

              "Structured Error Handling Example")

   * Show Unhandled exception with error inside CATCH

   x = laArrayNotDefined[100]

FINALLY

   * This code is optional and always runs

   MESSAGEBOX("Finally - made it!", 0+64, "Structured Error Handling Example")

* Unhandled exceptions are dealt with after the ENDTRY

ENDTRY

RETURN 

******************
PROCEDURE ErrHand

PARAMETER tnError, tcMessage, tcMessage1, ;

          tcProg, tnLineNo, tcMsgParameter

LOCAL lcMessage

lcMessage = "Global ErrorHandler:" + ccCRLF + ;

            tcMessage + " (" + ;

            TRANSFORM(tnError) + ")" + ccCRLF + ;

            "Code: " + ;

            ALLTRIM(tcMessage1) + ccCRLF + ccCRLF + ;

            "On line " + TRANSFORM(tnLineNo) + " of " + ;

            ALLTRIM(tcProg)

MESSAGEBOX(lcMessage, 0 + 48, "Global Error Handler")

RETURN
There are a few more differences with structured error handling you need to be aware of when implementing this error handler. First is the TRY…CATCH does not allow any RETRY or RETURN commands. It is fairly common to retry the error triggering code in a global or object-level error handler. Basically you will trigger another error: “RETURN/RETRY statement not allowed in TRY/CATCH” (2060).
Instead of passing in parameters to the error handler, the structured error handler works with an Exception object. The exception object is automatically created for you when you use the TO <object variablename> in CATCH clause. They are also created when you use the THROW command discussed in a moment. The Exception object has several members in addition to the standard ones seen on most VFP base classes. These are detailed in Table 1.
Table 1. Properties of the Exception object.

	Property
	Description
	Similar to
	Data Type

	Details
	Additional details about an exception object that are related to information specified by the Message property.
	SYS(2018)
	Character

	ErrorNo
	The error number for an Exception object.
	ERROR() or 
AERROR[1]
	Integer

	LineContents
	The contents of the line that produced an exception object.
	MESSAGE(1)
	Character

	LineNo
	The line number of the statement that produced an exception object.
	LINENO()
	Integer

	Message
	The actual error message for an exception object.
	MESSAGE()
	Character

	Procedure
	The name of the procedure or method where an exception occurs.
	PROGRAM()
	Character

	StackLevel
	The stack level for the program in which the exception occurs.
	ASTACKINFO[1]
	Integer

	UserValue
	Specifies the value passed by the THROW statement used to generate an exception. You can use this value as an object reference.
	N/A
	Variant


You can create your own subclass of the exception object, but only in program (PRG) code. You cannot have the CATCH statement instantiate your own customized exception object (it uses the base VFP exception object only), but you can create your exception object in the CATCH block, and THROW an exception to the next highest error handler. Just like procedural code, you can use the ERROR command to trigger the global error handler, or in object-level error handling you can raise the Error event. Inside structured error handling you can THROW the exception to the next highest error handler. You are passing around an object from one error handler to the next. This could be another TRY…CATCH, or it could be to the Error event or global handler. 
One of the keys before throwing the exception to the next error handler up the chain is to understand the type of error handler you have next up the chain. For this you need the SYS(2410) function. This function returns zero through three (detailed in Table 2).
Table 2. Values returned by SYS(2410).

	Value
	Meaning

	0
	VFP’s default error handler

	1
	TRY...CATCH...FINALLY Structure

	2
	Error Event

	3
	ON ERROR command


With this function you can wrap the logic used to throw the error to the next higher error handler (if one exists).

lnNextHandler = SYS(2410)
DO CASE

   CASE lnNextHandler = 1

      THROW "Error happened here"

   CASE lnNextHandler = 2 OR lnNextHandler = 3

      ERROR "Error happened here"
   OTHERWISE 

      * Hot water because we are at VFP level

ENDCASE

The implementation of TRY…CATCH improved two previously untrappable errors. The first is errors you have in reports, and the second is the invalid DBC backlinks in tables. These errors were previously reported using the default VFP error reporting.
Error handling priorities

If you implement no ON ERROR, no Error event method code, and no TRY…CATCH…FINALLY, you will get the VFP default error message. This is definitely the wrong approach.

Using one style
It is possible to implement only one type of the error handlers. If you just implement a global error handler, the ON ERROR takes precedence until you clear it or execute another ON ERROR to replace the original. Most errors can be handled in VFP 9, but prior versions of VFP have errors (like those in reports) you cannot handle and will be addressed only by the default VFP error handler.

If you only implement code in the Error event method, only the errors triggered in the object can be captured and handled. Code outside of objects will have errors displayed by the default error handler. 

If you only implement TRY…CATCH you can only capture errors triggered inside of a TRY block. All other errors will be handled by the default error handler.
Mixing two styles

I think by now you probably see the advantages of selecting a mix of the different error handlers your application architecture. How do the priorities work when you pair each of the styles together?
The most common mixture with older code is ON ERROR and Error event methods. In this case the Error event takes precedence when an error occurs in object method code and ON ERROR takes precedence when an error occurs outside a class definition or an error occurs inside a class definition but outside method code (error is detected at the point of instantiating the object via the CREATEOBJECT() or NEWOBJECT()). Otherwise, Visual FoxPro displays the default system error message.

The code found in Listing 3 shows a program with an ON ERROR calling a procedure. This program also instantiates an object with an Error event The Error event handles some Word Automation errors and then delegates all other errors to the global error handler by calling it directly. Most errors inside the object are handled by the Error event code, but one is delegated to the global error handler. The call loWord.ThisMethodDoesNotExist() is directly handled by the global error handler because the code is outside of the object.

Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called ShowOnErrorWithErrorEventExample.prg. The class with the Error event code is cusWordAutomationMix in the ErrorEventExample.VCX.
Listing 3. This is an example of mixing the ON ERROR and Error method for error handling.

LOCAL loWord AS cusWordAutomation OF ErrorEventExample

#DEFINE ccCRLF   CHR(13) + CHR(10)

ON ERROR DO ErrHand WITH ;

          ERROR(), MESSAGE(), MESSAGE(1),;

          PROGRAM(), LINENO(), SYS(2018)

loWord = NEWOBJECT("cusWordAutomationMix", "ErrorEventExample")

loWord.OpenDocument("DoesNotExist.doc")

loWord.OpenDocument(FULLPATH(CURDIR()) + "SampleWordDoc.doc")

IF ISNULL(loWord.oDocument)

   * No document to manuipulate

ELSE

   loWord.PlugBookmark("DoesNotExist", "Does not matter")

   loWord.PlugBookMark("SessionName", "Best Practices Error Handling")

ENDIF

loWord.ThisMethodDoesNotExist()

loWord.DumpErrorLog()

MODIFY FILE (loWord.cLogOutputFileName) NOWAIT

loWord = .NULL.

RETURN

PROCEDURE ErrHand

PARAMETER tnError, tcMessage, tcMessage1, ;

          tcProg, tnLineNo, tcMsgParameter

LOCAL lcMessage

lcMessage = "Global ErrorHandler:" + ccCRLF + ;

            tcMessage + " (" + ;

            TRANSFORM(tnError) + ")" + ccCRLF + ;

            "Code: " + ;

            ALLTRIM(tcMessage1) + ccCRLF + ccCRLF + ;

            "On line " + TRANSFORM(tnLineNo) + " of " + ;

            ALLTRIM(tcProg)

MESSAGEBOX(lcMessage, 0 + 48, "Global Error Handler")

RETURN

Augmenting an application which is just using ON ERROR to take advantage of the TRY…CATCH…FINALLY syntax introduced in VFP 8 has some straight forward precedence. Any errors occurring in a TRY block are handled by the CATCH. The ON ERROR code takes precedence on any errors outside of the TRY block including errors in the CATCH and FINALLY blocks. Otherwise the VFP default system error message is displayed.
The last of the paired combinations is the Error event with TRY…CATCH…FINALLY. This seems like an unlikely combination because ON ERROR code has been around since the beginning of time and TRY…CATCH has only been around since VFP 8. It is the most complicated of the paired scenarios. TRY…CATCH take precedence if any errors happen in a TRY block outside of the class, inside TRY block in a method in a class, an error is triggered when the class is instantiated within a TRY block, an error occurs in a TRY block inside external procedure or object method code, and the method or procedure is called directly or from another method, regardless of whether the method call appears inside or outside a TRY block. The Error event takes precedence if an error occurs in the object method code outside of a TRY block, or an error occurs in method code and the method code is called directly or from another method regardless if the method call is inside or outside of a TRY block.
Mixing all three styles
The next question is: what is the priority of how Visual FoxPro handles errors when all three types are integrated into the code? 

1. If one exists, the immediate TRY...CATCH in the same method/procedure that the error occurs. (Also remember this applies also to external procedures the code calls within the TRY)
2. The object’s Error event, if code exists, for the object
3. TRY...CATCH at the next level up in the call stack or in a higher-level method
4. ON ERROR routine, if one exists
5. The standard Visual FoxPro system error message
Error handling strategy

The most critical part of designing your error strategy is classifying the errors you want to deal with, and then determining where and how these errors are handled. The simpler the error handling strategy can be the better it will be in the long run. 
There are a number of approaches available including a mix of the three styles of error handling inside of Visual FoxPro. The approach you take will depend on the framework you have developed or purchased. It will also depend on the type of applications you develop.
Doug Hennig writes about a strategy I know a lot of developers have adopted and one I feel is a best practice. This strategy is quite simple, yet very elegant in implementation. Part of this strategy is detailed in his conference session Error Handling in VFP 8 and the other half is detailed in his online whitepaper called Error Handling in Visual FoxPro available online at http://stonefield.com.

To summarize the strategy, you use all three layers of error handling. TRY structures for local error handling as needed, Error event method code for encapsulated error handling, and the rest of the errors are captured in by the global error handler. This should be pretty obvious based on the overview of the three types of error handling discussed previously in this whitepaper.
The next recommendation is to wrap all your error handling code (in your Error event methods and the global error handling code) with TRY structures. If you do this your users never see the default VFP error handling message box when something is wrong with your error handling code. Remember though, error handling code has to be the best tested code you have.

Don’t use the THROW command unless you know there is a TRY structure is there to CATCH it. One other thing with TRY…CATCH, don’t wrap your entire application with one TRY structure. If you do, you don’t have a way to stay within the application if an error occurs.
The Error event method code is based on the Chain of Responsibility design pattern. Each Error event for each object which has code encapsulated to address errors for the object and has the default OTHERWISE behavior calls the superclass’s Error event method code and passes along the error parameters passed to the event in the first place. Each superclass has the opportunity to address the error or keep passing it along to the next superclass. The error code at the top of the class hierarchy can address the error, or it passes the error on to the container. Once it gets to the container it follows the same philosophy (it is just a different object) up the superclass hierarchy. Once you run out of containers the error parameters are passed to the global error handler.
The global error handler is an object that is instance to a global memory variable during the application startup. It has a simple programming interface because it deals with calls by the Error event methods and via the ON ERROR, and returns the choice of the user so it can be returned to the original calling object or program and the result can be handled. This code also performs the logging of the error if fatal (unhandled), can optionally bring up the debugger, or shut the application down.
My recommendation is to download Doug’s whitepaper and look at all his code. It is the best code I have seen with respect to error handling and I have learned a ton from studying it. I believe you will too.
Logging errors
Errors happening in production cannot be fixed unless the developer can track down the problem and apply the proper changes to the offending code. I believe you and I can agree the errors are easier to track down when we know the exact problem and were in the code the error is triggered. Logging the errors and key information about the error and the application environment is going to provide the fastest path to squashing the problem. What are the best practices to logging the errors and the environment?
Who, when, where, why, and how

There are some basics all error handlers need to log (see Table 3). This information is going to be critical to understand if you are researching the error the user reported to you. It provides you the basis to looking at the code with the problem.

Minimally you want to track the error number, the error message, the date and time the error occurred, who the user is, the computer running the program, the line number, the program or method with the error, and the line of code if you know it (only if the Debug Code is compiled into the executable).

Table 3. These are the base items you want to log and how you get the details to track.

	Item
	Description

	Error Number
	ERROR(), usually passed into error handler, available by default in the Error event as a parameter, and ErrorNo property on an exception object.

	Error Message
	MESSAGE(), usually passed into error handler, not available for Error event directly but can use AERROR(),and Message property on an exception object.

	Program
	PROGRAM(), usually passed into error handler, available by default in the Error event as a parameter, and Procedure property on an exception object.

	Line Number
	LINENO(),usually passed into error handler, available by default in the Error event as a parameter, and Procedure property on an exception object, but only available if you have debug code compiled into the application or are running in the VFP IDE.

	Line of code
	MESSAGE(1) if included on the ON ERROR line as a parameter, not available for Error event directly but can use AERROR(), and LineContents property on an exception object.

	Date and time
	DATETME(), or DATE() and TIME() if you are doing two fields

	User
	Depends on the framework or your login scheme, you can also use SYS(0) to get the network login and computer name.


The reason for logging this information is to get a start on the fix. The details are simple to retrieve and do not call any user defined functions. One of the problems with calling a user defined function from your error handler is dealing with the possibility you have already exceeded the call stack limitation. This minimalist approach avoids this problem.
Use VFP’s LIST command
The LIST command has many options to capture settings, values of variables, connections, tables, views, and environment information. The LIST command (see example in Listing 4) can dump the details to the screen, printer, or into a file (important for Error logging). 
Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called ErrorHandlerListCommands.prg, which includes the code to capture the important LIST command details.

Listing 4. One possible approach to capturing the information provided by the VFP LIST command.
lcBaseFileName = ADDBS(SYS(2023)) + "ERR" + DTOS(DATE())+SYS(2015)
* ------  Complete Memory Dump  ------

lcErrorFileName = FORCEEXT(lcBaseFileName + "M", "err")
LIST MEMORY TO FILE (lcErrorFileName) NOCONSOLE

* ------  Complete Status Dump  ------

lcErrorFileName = FORCEEXT(lcBaseFileName + "S", "err")

LIST STATUS TO FILE (lcErrorFileName) NOCONSOLE 

* ------  Complete Objects Dump  ------

lcErrorFileName = FORCEEXT(lcBaseFileName + "O", "err")

LIST OBJECTS TO FILE (lcErrorFileName) NOCONSOLE 

* ------  Complete Connections Dump  ------

lcErrorFileName = FORCEEXT(lcBaseFileName + "C", "err")

LIST CONNECTIONS TO FILE (lcErrorFileName) NOCONSOLE 

* ------  Complete DLL Dump  ------

lcErrorFileName = FORCEEXT(lcBaseFileName + "D", "err")

LIST DLLS TO FILE (lcErrorFileName) NOCONSOLE 

The files can be pulled into memo fields or left on the drive for analysis by a developer or tech support technician. One issue to consider is the LIST MEMORY and an impact of security of the application. Since the LIST MEMORY dumps all the variables and their values, you could be revealing secured, but unencrypted data in the dump (see Figure 3). If this is an issue for you or your customers you might have to scan the file for some keywords or variable names and wipe out the secure information before pulling in the dump into the final logged state.

Figure 3. The LIST MEMORY command will include secure data and could possibly serve as a security breach with respect to sensitive data.

The LIST MEMORY command may be the most powerful of the commands with respect to error handling. It is able to see all the variables in the calling stack, not just the ones currently in scope. This allows you to see variable values in procedures up the call stack, how they are scoped, and the data type.

The LIST STATUS only sees the settings for the current datasession. If you need to see the same details you need to loop through all the open datasessions and collect the information yourself.
AERROR function array contents
The AERROR function collects details about the error and places them into an array. This array contains numeric and character data, and frequently will have NULL data. It all depends on the type of error and what information is available for Visual FoxPro to capture. The trick to logging is generically convert the array to a character string using code similar to what is found in Listing 5:
Listing 5. The code collects all the information from the AERROR array and dumps it out to a file so you can later append it into a memo field in the error log.
LOCAL laError, ;

      lcAError, ;

      lcString, ;

      lnIndex

#DEFINE ccCRLF            CHR(13)+CHR(10)

#DEFINE ccNULL_LANGUAGE   "<Unknown>"

DIMENSION laError[1]

AERROR(laError)

lcAError = "AERROR() contents collected" + ccCRLF

lcAError = lcAError + ;

           '[1] = ' + ;

           ALLTRIM(STR(laError[1])) + ;

           ccCRLF

lcAError = lcAError + ;

           '[2] = ' + ;

           laError[2] + ;

           ccCRLF

lcAError = lcAError + ;

           '[3] = ' + ;

           NVL(laError[3],ccNULL_LANGUAGE) + ;

           ccCRLF

FOR lnIndex = 4 TO 7

   luValue = laError[lnIndex]

   DO CASE

      CASE ISNULL(luValue)

         lcString = ccNULL_LANGUAGE

      CASE TYPE("luValue") = "C"

         lcString = luValue

      CASE TYPE("luValue") = "N"

         lcString = ALLTRIM(STR(luValue))

      OTHERWISE

         lcString = "Unable to determine"

   ENDCASE

   lcAError = lcAError + ;

              '[' + ALLTRIM(STR(lnIndex)) + '] = ' + ;

              lcString + ;

              ccCRLF

ENDFOR

* ------  Complete AERROR Dump  ------

lcBaseFileName  = ADDBS(SYS(2023)) + "ERR" + DTOS(DATE())+SYS(2015)

lcErrorFileName = FORCEEXT(lcBaseFileName + "A", "err")

STRTOFILE(lcAError, lcErrorFileName, 0)

RETURN lcErrorFileName
Environment settings

I find it important to understand the Visual FoxPro environment and the specification for the workstation the app is running on, or in this case is triggering the error. The more I know about the machine and the state of the VFP settings for things like pathing, memory, drive space, printers, and operating systems, the easier I solve problems. The program I use to collect this information is detailed in Listing 6.
Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called CollectEnvironmentDetails.prg.
Listing 6. Code to collect environment settings and store them in a file.

LOCAL lcBaseFileName, ;

      lcErrorFileName, ;

      lnPadTabs, ;

      lcOldConsole

* ------  Complete System Error Dump  ------

lcBaseFileName  = ADDBS(SYS(2023)) + "ERR" + DTOS(DATE())+SYS(2015)

lcErrorFileName = FORCEEXT(lcBaseFileName + "E", "err")

lnPadTabs       = 30

lcOldConsole    = SET("CONSOLE")

SET CONSOLE OFF

SET ALTERNATE TO (lcErrorFileName)

SET ALTERNATE ON

? "********* Pathing Settings *************"

? PADR("SET PATH", lnPadTabs) + ": " + LOWER(SET("PATH"))

? PADR("SET CLASSLIB", lnPadTabs) + ": " + LOWER(SET("CLASSLIB"))

? PADR("SET PROCEDURE", lnPadTabs) + ": " + LOWER(SET("PROCEDURE"))

? PADR("SET LIBRARY", lnPadTabs) + ": " + LOWER(SET("LIBRARY"))

?

? "******Visual FoxPro Settings ***********"

? PADR("FoxPro Serial Number", lnPadTabs) + ": " + SYS(9)

? PADR("FoxPro Version", lnPadTabs) + ": " + VERSION(1)

? PADR("FoxPro Dev or Run-time", lnPadTabs) + ": " + ;

       IIF(VERSION(2)=0, "Run-time", "IDE")

? PADR("FoxPro Localized Language", lnPadTabs) + ": " + VERSION(3)

? PADR("FoxPro Started", lnPadTabs) + ": " + LOWER(HOME(0))

? PADR("FoxPro Root folder", lnPadTabs) + ": " + LOWER(HOME(1))

? PADR("User Application Data folder", lnPadTabs) + ": " + LOWER(HOME(7))

? PADR("Default FoxPro Printer", lnPadTabs) + ": " + SET("Printer", 2)

? PADR("Default OS Printer", lnPadTabs) + ": " + SET("Printer", 2)

? PADR("Printer Status", lnPadTabs) + ": " + SYS(13) 

? PADR("Printer Device", lnPadTabs) + ": " + SYS(6)

? PADR("Current Lock Status", lnPadTabs) + ": " + SYS(2011) 

? PADR("Memo Field Blocksize", lnPadTabs) + ": " + SYS(2012) 

? PADR("Config.FPW File", lnPadTabs) + ": " + LOWER(FULLPATH(SYS(2019)))

IF VERSION(5) >= 900

   ? PADR("Config.FPW Internal File", lnPadTabs) + ": " + LOWER(SYS(2019, 2))

ENDIF

? PADR("Resource File", lnPadTabs) + ": " + LOWER(SYS(2005))

? PADR("Current CONSOLE setting", lnPadTabs) + ": " + lcOldConsole

? PADR("Current DEVICE setting", lnPadTabs) + ": " + SYS(101) 

? PADR("Current FORMAT file", lnPadTabs) + ": " + SYS(7)

? PADR("Current TALK setting", lnPadTabs) + ": " + SYS(103)

? PADR("Turn Cursor on or off", lnPadTabs) + ": " + SYS(2002)

? PADR("Julian System Date", lnPadTabs) + ": " + SYS(1)

? PADR("Seconds Since Midnight", lnPadTabs) + ": " + SYS(2)

? PADR("Help ON or OFF", lnPadTabs) + ": " + SET("Help")

? PADR("Help File", lnPadTabs) + ": " + SET("Help",1)

? 

? "********** RAM Usage *******************"

? PADR("Processor in use", lnPadTabs) + ": " + SYS(17) 

? PADR("Remaining memory below 640K", lnPadTabs) + ": " + SYS(12) 

? PADR("FoxPro EMS memory usage", lnPadTabs) + ": " + SYS(23) 

? PADR("EMS memory limit", lnPadTabs) + ": " + SYS(24) 

? PADR("FoxPro's available memory", lnPadTabs) + ": " + SYS(1001) 

? PADR("User object memory use", lnPadTabs) + ": " + SYS(1016) 

? PADR("Memory for RUN command", lnPadTabs) + ": " + SYS(12)

? PADR("Buffer foreground memory", lnPadTabs) + ": " + TRANSFORM(SYS(3050,1))

? PADR("Buffer background memory", lnPadTabs) + ": " + TRANSFORM(SYS(3050,2))

?

? "***** Drives and Disk Usage ************"

? PADR("DEFAULT drive", lnPadTabs) + ": " + SYS(5) 

? PADR("Current directory", lnPadTabs) + ": " + LOWER(FULLPATH(CURDIR()))

? PADR("Current resource file", lnPadTabs) + ": " + LOWER(SYS(2005))

? PADR("CONFIG.SYS file setting", lnPadTabs) + ": " + SYS(2010) 

? PADR("Disk free space", lnPadTabs) + ": " + ;

        TRANSFORM(VAL(SYS(2020)),"999,999,999,999,999")

? PADR("Disk size", lnPadTabs) + ": " + ;

        TRANSFORM(DISKSPACE(JUSTDRIVE(CURDIR()), 1), "999,999,999,999,999")

? PADR("Percentage free", lnPadTabs) + ": " + ;

        TRANSFORM((VAL(SYS(2020)) / DISKSPACE(JUSTDRIVE(CURDIR()), 1)) * 100) ;

        + "%"

? PADR("Disk cluster size", lnPadTabs) + ": " + SYS(2022)

? PADR("Temporary file path", lnPadTabs) + ": " + LOWER(SYS(2023))

?

? "********* System Settings **************"

? PADR("Operating System", lnPadTabs) + ": " + OS(1)+ "." +  OS(5)

? PADR("Underlying Operating System", lnPadTabs) + ": " + OS()

? PADR("Operating System Service Pack", lnPadTabs) + ": " + OS(7)

? PADR("Operating System Platform", lnPadTabs) + ": " + OS(6)

? PADR("Processor", lnPadTabs) + ": " + SYS(17)

? PADR("Graphics Card", lnPadTabs) + ": " + SYS(2006)

? PADR("Files In CONFIG.SYS", lnPadTabs) + ": " + SYS(2010)

? PADR("Network machine # Login", lnPadTabs) + ": " + SYS(0)

?

? "*: EOF :*"

SET CONSOLE &lcOldConsole

SET ALTERNATE OFF

SET ALTERNATE TO 

RETURN lcErrorFileName

Program call stack
You need to understand the path the programs took to get to the offending code. This is where the program call stack comes in handy. For years we used SYS(16) in a loop, but now we have ASTACKINFO(), which was included in VFP 7.0 and provides more details about the call stack than SYS(16). You can review the  program to extract details about the call stack in Listing 7.
Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called CollectCallStackDetails.prg.

Listing 7. Code to collect call stack details and store them in a file.

LOCAL lcCallStack, ;

      lcCallStack, ;

      lcString, ;

      lnIndex, ;

      lnStackCount, ;

      lcBaseFileName

#DEFINE ccCRLF            CHR(13)+CHR(10)

#DEFINE ccTAB             CHR(9)

#DEFINE ccSPACE           SPACE(1)

LOCAL ARRAY laStackInfo[1]

ASTACKINFO(laStackInfo)

lnStackCount = ALEN(laStackInfo, 1)

lcCallStack  = SPACE(0)

IF lnStackCount > 0

   * Process the call stack from the latest to the first

   * You might want to remove calls to the error handler

   FOR lnIndex =  TO 1 STEP -1

      lcCallStack = lcCallStack + ; 

                    '[' + TRANSFORM(laStackInfo[lnIndex,1]) + '] = ' + ;

                    laStackInfo[lnIndex,3] + ;

                    " of " + ;

                    laStackInfo[lnIndex,2] + ;

                    ccCRLF + ccTAB + ;

                    "Source is " + ;

                    laStackInfo[lnIndex,4] + ;

                    ccCRLF + ccTAB + ;

                    "Called on line " + ; 

                    TRANSFORM(laStackInfo[lnIndex,5]) + ;

                    " " + ;

                    ccCRLF + ccTAB + ;

                    ALLTRIM(laStackInfo[lnIndex,6],0,CHR(32), ccTAB) + ;

                    ccCRLF

   ENDFOR

ELSE

   * Should never happen if a program is running

   lcCallStack = "Weird, no call stack to report."

ENDIF

* ------  Complete ASTACKINFO Dump  ------

lcBaseFileName  = ADDBS(SYS(2023)) + "ERR" + DTOS(DATE())+SYS(2015)

lcErrorFileName = FORCEEXT(lcBaseFileName + "L", "err")

STRTOFILE(lcCallStack, lcErrorFileName, 0)

RETURN lcErrorFileName

Form datasessions

Errors are often related to the condition of the datasession, the state of the cursors (edit records, unedited records), end of file, beginning of file, or the values contained in the records of the cursors. At the same time your applications can have numerous forms opened in the default datasession or in multiple private data sessions. It is a good idea to gain an understanding of the current state of the data when the program crashes. It is very helpful when debugging data related errors in your applications.
One of the smartest developers we had in the FoxPro Community is the late Drew Speedie. Drew is the original architect of Visual MaxFrame Professional. I had the opportunity to work with this framework many years ago and learned a lot about error handling from Drew’s code. In particular, Drew has some great code when it comes to logging the state of the data. Over the years I migrated this code to other projects I have worked on and improved the code to meet my needs, but the original code is based on Drew’s ideas. You can see example output in Listing 8.
In the interest of space I am not going to include the code in this whitepaper, but you can review it by looking at the CollectFormDataSessionDetails.PRG included with the session downloads. What is important about this code is it captures information about the current active form first (if a form is active), and then looks at all the other forms. The reason for this is simple, the program likely crashed when you are using the active form. 
Listing 8. This is example output logged by a program dedicated to understanding the state of the different datasessions in used by the application.

CURRENT ACTIVE FORM

               Name  v_instrumentsubcategoryform1

            Caption  Instrument Subcategory:1

           SCX file  (VCX or PRG based form)

              Class  V_instrumentsubcategoryform

       ClassLibrary  j:\wlcproject\CTS\CHA\libs\ainstrumentsubcategory.vcx

        DataSession  Private

      DataSessionID  5

  Transaction level  0

    SET DATABASE TO  vCHA

      Current DBC()  j:\wlcproject\CTS\CHA\data\vCHA.dbc

         WORK AREAS  

            1) v_instrumentsubcategory

                     DBF()  c:\temp\00007oqx01oh.tmp

                   RECNO()  eof()

                RECCOUNT()  eof()

                     EOF()  yes

                   ORDER()  none

                  FILTER()  none

                  DATABASE  j:\wlcproject\CTS\CHA\data\vCHA.dbc

                SourceType  2

               SendUpdates  .T.

          Connection Handle 3

                       SQL  SELECT InstrumentSubCategory.cInstrumentsubcategory_pk, InstrumentSubCategory.cCode, InstrumentSubCategory.cDescription, InstrumentSubCategory.lActive, InstrumentSubCategory.tEntry, InstrumentSubCategory.cLastaction, InstrumentSubCategory.cUserid, InstrumentSubCategory.tUpdated FROM dbo.InstrumentSubCategory WHERE cDescription LIKE ?vp_cDescription ORDER BY cDescription 

                    Tables  dbo.InstrumentSubCategory

         Update Field List  cinstrumentsubcategory_pk, ccode, cdescription, lactive, tentry, clastaction, cuserid, tupdated

            Parameter List  vp_cDescription,'C'

                SourceName  v_instrumentsubcategory

                BufferMode  5

           GetFldState(-1) is at EOF()

OTHER OPEN FORMS

** Form 1**

               Name  dbcxmgr

            Caption  

           SCX file  (VCX or PRG based form)

              Class  Dbcxmgr

       ClassLibrary  j:\wlcproject\framework\vfe8\vfeframe\libs\dbcxmgr.vcx

        DataSession  Private

      DataSessionID  2

  Transaction level  0

    SET DATABASE TO  none

      Current DBC()  none

         WORK AREAS  

            1) dbcxprops

                     DBF()  c:\temp\00007oqx01hx.tmp

                   RECNO()  57

                RECCOUNT()  122

                     EOF()  no

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  c:\temp\00007oqx01hx.tmp

                BufferMode  1

            2) sdtuser

                     DBF()  j:\wlcproject\CTS\CHA\metadata\sdtvuser.dbf

                   RECNO()  eof()

                RECCOUNT()  eof()

                     EOF()  yes

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  j:\wlcproject\CTS\CHA\metadata\sdtvuser.dbf

                BufferMode  1

            3) sdtmeta

                     DBF()  j:\wlcproject\CTS\CHA\metadata\sdtvmeta.dbf

                   RECNO()  eof()

                RECCOUNT()  eof()

                     EOF()  yes

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  j:\wlcproject\CTS\CHA\metadata\sdtvmeta.dbf

                BufferMode  1

            4) vfemeta

                     DBF()  j:\wlcproject\CTS\CHA\metadata\vfemeta.dbf

                   RECNO()  229

                RECCOUNT()  561

                     EOF()  no

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  j:\wlcproject\CTS\CHA\metadata\vfemeta.dbf

                BufferMode  1

            5) coremeta

                     DBF()  j:\wlcproject\CTS\CHA\metadata\coremeta.dbf

                   RECNO()  eof()

                RECCOUNT()  eof()

                     EOF()  yes

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  j:\wlcproject\CTS\CHA\metadata\coremeta.dbf

                BufferMode  1

            6) dbcxreg

                     DBF()  j:\wlcproject\CTS\CHA\metadata\dbcxreg.dbf

                   RECNO()  eof()

                RECCOUNT()  eof()

                     EOF()  yes

                   ORDER()  none

                  FILTER()  none

                  DATABASE  free table

                SourceType  3

                SourceName  j:\wlcproject\CTS\CHA\metadata\dbcxreg.dbf

                BufferMode  1

There is definitely some room for improvement in the example code. The program does not take into account special needs of CursorAdapters. I am not using CursorAdapters in my application development so I have not modified the code to address this issue.
Config.FPW contents

The last logging detail (Listing 9) I am going to address is recording the contents of the Config.FPW file. You might ask the question: why would you record the Config.FPW file if every user uses the same file? 
The reason I record it is a simple one: There are many ways you can specify how Visual FoxPro determines the Config.FPW it uses. More importantly, there are many ways your users can surprise you by overriding the Config.FPW you thought you supplied with your application. Visual FoxPro 9 allows you to have an internal configuration file (compiled into your executable) and a secondary external configuration file. This means users can find ways of using a Config.FPW file your application was not necessarily expecting. They also can edit the external files and make settings your application might not find compatible. 
Included in the Developer Downloads for this whitepaper, available from the conference download site, is a program called CollectConfigFpwDetails.prg.

Listing 9. Code to collect Config.FPW details and store them in a file.

LOCAL lcString, ;

      lnIndex, ;

      lcBaseFileName

#DEFINE ccCRLF            CHR(13)+CHR(10)

#DEFINE ccNULL_LANGUAGE   "<Unknown>"

lcString = "Config.FPW File(s)" + ;

           ccCRLF + ccCRLF + ;

           "External File: "

IF !EMPTY(SYS(2019)) AND FILE(SYS(2019))

   lcString = lcString + ;

              LOWER(SYS(2019)) + ccCRLF + ccCRLF + ;

              FILETOSTR(SYS(2019)) + ;

              ccCRLF + ccCRLF

ELSE

   lcString = lcString + ;

              "none" + ccCRLF + ccCRLF

ENDIF

IF VERSION(5) >= 900

   lcString = lcString + ;

              "Internal File: "

   IF !EMPTY(SYS(2019, 2)) AND FILE(SYS(2019, 2))

      lcString = lcString + ;

                 LOWER(SYS(2019, 2)) + ccCRLF + ccCRLF + ;

                 FILETOSTR(SYS(2019))

   ELSE

      lcString = lcString + ;

                 "none" + ccCRLF

   ENDIF

ENDIF

* ------  Complete Config.FPW Dump  ------

lcBaseFileName  = ADDBS(SYS(2023)) + "ERR" + DTOS(DATE())+SYS(2015)

lcErrorFileName = FORCEEXT(lcBaseFileName + "W", "err")

STRTOFILE(lcString, lcErrorFileName, 0)

RETURN lcErrorFileName

Communicating errors and bugs
A perfect world would not have applications released with code that crashes, but the complexity of software today sometimes makes it almost impossible to do so. Some customers are shy about reporting what they think are bugs because they feel they have done something wrong to break the program. Others are going to pull your chain every time they find something wrong. 

Tracking issues assists you in a number of ways. First, it betters the software you deliver to the customers, which in turn improves customer relations. Second, it provides you a list of things to work on for the various clients. Third, it provides the developers with a training mechanism for better development and testing techniques.

Fatal error bugs, the unexpected ones or the ones your customers find ways of producing, are fairly easy to track as we have discussed throughout this whitepaper. It is important to provide a mechanism for the users to transmit the collected problems. Options include a report that formats the information collected. Another option is to print the report to PDF format and attach the PDF file to an e-mail and have it sent to your support e-mail address. The most recent idea we have is to transform the error information tracked in the error table into XML and have the XML sent via e-mail. The advantage of this method of transfer allows you to import the information directly into the support database.

It is important the user specifies a number of key elements. The elements I like to have reported are:

· Exact steps to reproduce the error

· What was observed

· What was expected

· Additional comments from the user

· Details we logged when the program crashed

So what do you do for the non-fatal errors? This is the kind of bug the user reports that allegedly does not meet the requirements or when the application fails to operate as they expect. The same mechanism to transport fatal errors can be used to transport user reported issues.
Phone Call
You and I have received the dreaded phone call when the user starts out a conversation with “your program crashed.” No one likes to hear about crashes and the high tension discussion on data loss, hours of processing, and the fact work is halted in the customer office. 
At the same time it makes it almost impossible for developers to figure out the problem without understanding the steps taken to reproduce the problem. I don’t know about you, but most phone calls detailing errors usually go something like this:
“Hi Rick, your program crashed when I ran a report.”

“Hello sir, which report?”

“You know the one with the order information.”

“There are several reports with order information, can you be more specific?”

“The one I run every day at 3:00, just before I am ready to leave for the day.”

“Oh, that report! Refresh my foggy mind, what menu item did you pick to run the report?”

Users have a way of simplifying the steps it takes to reproduce the problem within the context of their job. This might be good and it might be bad. Optimized steps helps narrow down the problem faster, but often I find they skip the automatic stuff often critical in determining the problem. I don’t know about everyone else, but phone calls should be limited to the notification of the problem and used to initiate the influx of details recorded by the error handler.
Screen shots

I often request the user to reproduce the error and send me a screen shot. This provides two benefits: one is the user has to retrace their steps and understand how they got to the error and the second is giving me a visual of the application and where they are when the problem happens.
The problem with a screen shot is it tells only part of the story and not all errors can be shown in a screen shot. Another problem is the screen shots sizes often exceed one megabyte. This is not normally a problem with most of you having broadband access, but occasionally we are disconnected or have to access the Internet via a dial up connection. 
Recording a screencast

One of my favorite and newer ways of receiving error reports is a screencast. Just in case you are not familiar with this technology, I am talking about a recorded video made by the user. I have a couple of customers who use Camtasia from TechSmith (http://techsmith.com/). This is way better than a screen shot because you can see and hear the user step through the application. It reveals the reproducible steps in a concise and understandable way. I also can record screencasts to show users the workaround to an error in addition to demonstrating a fix in the works.
Screencast videos put screen shots to shame, but some users are understandably uncomfortable putting them together. They are large files too so broadband connections are a must on both sides. The screencast also suffers from the fact all the error handling logged details are not sent, but the user can open the error log form in the application and show you the various items recorded. It is a terrific supplement to the details in the error log.
Send the error log files

The most common way for users to communicate errors back to the tech support folks is to zip up the error log files and send them via email or upload to a FTP site. This is only possible if you have your error log files in DBF or text format. Developers who have their error log recorded in a backend database like SQL Server, Oracle, etc. need an export mechanism before the files can be sent.
The best part about this type of communication is you have all the details available in the log and the most information to work with. You also have the possibility to transfer the error information to your bug tracking system through an automated process. It is not unusual for me to connect to the client via PcAnywhere to check out the error log when a problem occurs and easily can email the files back to my office for a detailed analysis or pull them via file transfer.
Send a report

One of the more common ways of reporting errors is running a report on the logged details and sending this as an Acrobat PDF attachment in email. I discourage FAX usage unless there are absolutely no alternatives. Depending on the amount of information you log you might want to develop both a short form and an all inclusive form for this type of reporting. Users can easily run the report (Figure 4) and send it to the tech support technician.

Figure 4. Printing a long error report provides the user one mechanism to deliver the logged error details to the tech support technician.

The process of generating a report to PDF and emailing it is out of the scope of this session, but there are plenty of resources to accomplish this including MegaFox: 1002 Things You Wanted To Know About Extending Visual FoxPro available from Hentzenwerke Publishing, the Fox Wiki (http://fox.wikis.com), and Craig Boyd’s blog (http://sweetpotatosoftware.com/blog).
RSS feed
A couple of years ago I had an idea of sending the error log information via XML, but did not pursue it to the point of implementing it with my error logging features of the applications I deliver to customers. In late 2004 I started working with Really Simple Syndication (RSS) files every day to keep up on news and blogs. A thought occurred to me about using RSS as a file transmission of error logs back to me (Listing 10). This thought was placed on the back burner because it was a low priority. In October 2005 I attended Rick Borup’s session titled Integrating RSS with Visual FoxPro Applications at the Southwest Fox conference. During this session Rick explained how he is using RSS files to accomplish this exact task. So instead of duplicating much of Rick’s work I asked his permission to use the concepts and code he established and has published on his Web site.

Rick Borup from Information Technology Associates has an excellent whitepaper on RSS on his company Web site (http://ita-software.com/papers/borup_rss.pdf).
The key to the concept of RSS is to understand the layout of an RSS file, which is straight XML, but is expected to conform to the RSS 2.0 specification. Rick’s whitepaper on this subject covers all the RSS basics, the file structure, the validation levels of the file, and how you can verify the file format is valid. 
Listing 10. The following code processes the ErrorLog table and outputs a RSS 2.0 file using Rick Borup’s MyRSSHandler class provided in the conference downloads.

LPARAMETERS tcFileName, tlAllRecords

LOCAL loRSS, ;

      lcFileName, ;

      lcScanScope, ;

      llOpenedErrorLog, ;

      lcOldSafety

* Better handled by data driven parameters

#DEFINE ccAUTHOR "raschummer@whitelightcomputing.com (Rick Schummer)"

*< SET STEP ON 

lcOldSafety = SET("Safety")

SET SAFETY OFF

* Set procedures needed for this process

SET PROCEDURE TO RFC822Date, ;

                 GetNewGUID ;

    ADDITIVE 

* Handle missing parameters

IF PCOUNT() < 1 OR VARTYPE(tcFileName) # "C"

   tcFileName = "MyErrorLog.rss"

ELSE

   tcFileName = FORCEEXT(ALLTRIM(tcFileName), "rss")

ENDIF

IF tlAllRecords

   lcScanScope = SPACE(0)

ELSE

   lcScanScope = "NEXT 1"

ENDIF

* Initialize/setup variables

lcFileName = "http://localhost/ConferenceExamples/" + tcFileName

* Register DLLs for GUID Generation

InitializeGUIDGenerator()

loRSS = NEWOBJECT( "RSSHandler", "myRSSHandler.prg")

loRSS.InitializeFeed( ;

      "Application Error Log", ;

      lcFileName, ;

      "Conference Example Error Log RSS Feed", ;

      RFC822Date( DATETIME()) ;

     )

IF NOT USED("ErrorLog")

   USE ErrorLog IN 0 AGAIN SHARED NOUPDATE ALIAS ErrorLog

   llOpenedErrorLog = .T.

ENDIF

SELECT ErrorLog

SCAN &lcScanScope

  loRSS.InsertItem( ;

     "Error log entry " + TTOC(ErrorLog.tDateTime), ;

     "<![CDATA[" + ;

       "<b>User:</b> " + ALLTRIM(ErrorLog.cUserID) + "<br />" + CHR(13) + ;

       "<b>Error:</b> " + TRANSFORM(ErrorLog.nError) + "<br />" + CHR(13) + ;

       "<b>Message:</b> " + ALLTRIM(ErrorLog.cMessage) + "<br />" + CHR(13) + ;

       "<b>Line:</b> " + TRANSFORM(ErrorLog.nLineNo) + "<br />" + CHR(13) + ;

       "<b>Source:</b> " + ProcessMemo(ErrorLog.mSource) + "<br />" + ;

                           CHR(13) + ; 

       "<b>Method:</b> " + ALLTRIM(ErrorLog.cMethod) + "<br />" + CHR(13) + ; 

       "<b>Call Stack:</b> " + ProcessMemo(ErrorLog.mCallStack) + "<br />" + ;

                               CHR(13) + ;

       "<b>Memory:</b> " + ProcessMemo(ErrorLog.mMemory) + "<br />" + ;

                           CHR(13) + ;

       "<b>mStatus:</b> " + ProcessMemo(ErrorLog.mStatus) + ;

    "]]>", ;

    lcFileName, ;

    ccAUTHOR, ;

    SUBSTRC(GetGUID(), 2, 36), ;

    RFC822Date(DATETIME()) ;

   )

ENDSCAN

loRSS.TerminateFeed()

loRSS.WriteFile(JUSTFNAME(lcFileName))

IF llOpenedErrorLog

   USE IN SELECT("ErrorLog")

ENDIF

SET SAFETY &lcOldSafety

RETURN

**********************************

PROCEDURE ProcessMemo(tcMemoField)

LOCAL lcReturnVal

IF PCOUNT() < 1 OR VARTYPE(tcMemoField) # "C"

   lcReturnVal = SPACE(0)

ELSE

   tcMemoField = ALLTRIM(tcMemoField)

   * Add HTML line breaks where there are carriage returns and linefeed 

   lcReturnVal = STRTRAN(tcMemoField, CHR(13), "<br />" + CHR(13))

ENDIF

RETURN lcReturnVal

The user can transmit this file to you via email, or can post it to their Web server in a folder that is not searched by bots and search engines. Here is an example of the file for one error (with the memory and status dump removed to conserve on space):
<?xml version="1.0" encoding="utf-8"?>

<rss version="2.0">

<channel>

   <title>Application Error Log</title>

   <link>http://localhost/ConferenceExamples/CustomerErrorLog.rss</link>

   <description>Conference Example Error Log RSS Feed</description>

   <lastBuildDate>Tue, 07 Mar 2006 00:03:22 GMT</lastBuildDate>

   <item>

      <title>Error log entry 03/06/2006 07:03:22 PM</title>

      <description><![CDATA[<b>User:</b> RickSchummer<br />

<b>Error:</b> 1734<br />

<b>Message:</b> Property HASCHILDREN is not found.<br />

<b>Line:</b> 29<br />

<b>Source:</b> IF .HasChildren()<br />

<b>Method:</b> refreshchildren<br />

<b>Call Stack:</b> 000086BT005A<br />

BUGAPPLICATIONOBJECT.INIT<br />

BUGAPPLICATIONOBJECT.DO<br />

V_CSRCENTRALORDERS.GRDV_CSRCENTRALORDERS.GRCV_CSRCENTRALORDERS_CSELLERLASTNAME.TXTV_CSRCENTRALORDERS_CSELLERLASTNAME.DBLCLICK<br />

IGLOBALHOOK.DBLCLICKHOOK<br />

V_CSRCENTRALORDERS.GRDV_CSRCENTRALORDERS.DBLCLICK<br />

IGLOBALHOOK.DBLCLICKHOOK<br />

V_CSRCENTRALORDERS.GRDV_CSRCENTRALORDERS.SELECTRECORD<br />

V_CSRCENTRALORDERS.REFRESHUI<br />

V_CSRCENTRALORDERS.REFRESHCHILDREN<br />

V_CSRCENTRALORDERS.ERROR<br />

<br />

]]></description>

      <link>http://localhost/ConferenceExamples/CustomerErrorLog.rss</link>

      <author>raschummer@whitelightcomputing.com (Rick Schummer)</author>

      <guid isPermaLink="false">095BEAE8-1667-40F7-9650-876F60FBD241</guid>

      <pubDate>Tue, 07 Mar 2006 00:03:22 GMT</pubDate>

   </item>

</channel>

</rss>

If the user mails me the file I can copy it to a local folder and read the file in FeedDemon (RSS Aggregator/Reader from Newsgator). If they post it in their Web server I will see it the next time FeedDemon (Figure 5) does a pass.

Figure 5. Using a RSS aggregator/reader is recommended if you decide to output your error log to a RSS file.

I have just started using the RSS feeds and it is still at the experimental stage. In the future I see importing the data into a table and even creating bug reports directly in my bug reporting package.
Tracking errors
Back at the bat cave you need to have a mechanism to track reported bugs. There are a number of ways to handle this including pencil and paper, a Word document, or more likely a software package dedicated to tracking bugs. Doing a web search reveals more than a dozen different bug tracking solutions. The BugTrackingSoftware topic on the Fox Wiki links to more than a half dozen available products. Visual Studio 97 (with which VFP 5 was included) includes a solution called Anomaly Tracking System (ATS) written in Visual FoxPro (Figure 7). Visual Studio 6 includes a web version called ATSWeb. One nice thing about being database developers is you can augment the ATS products or build a solution that meets your needs if a satisfactory solution cannot be found. Some products like IssueView (Figure 8) from a company called IssueView.com has both a desktop component (based on a SQL Server or MS Access database) and an Internet component, which uses Active Server Pages with full, built-in e-mail notification.

There are other Web-based options that can be used including the new Customer Service Center from F1 Technologies (makers of Visual FoxExpress), BugCentral.com (based on VFP and WebConnect, see Figure 6), and Steven Black’s Wiki (another VFP and WebConnect implementation). The advantage of using a Web-based bug tracking application is your customers can directly enter in reports, geographically diverse development teams can access a common set of reported bugs, and customers can access reports to see the status of reports they made as well as reports from other users. 

Figure 6. BugCentral.com is one example of a pure Web-based bug tracking application.

Microsoft products automatically send statistics back to Microsoft via the Internet whenever a GPF or other serious error occurs via the Dr. Watson Error Reporting. Are we the only developers in the world that get some pleasure each time this happens and hoping some Microsoft Web server is getting bombarded with the same reports? There are many ways to get the reports programmatically via e-mail, a Web service, and faxing.

Figure 7. Some bug tracking packages like Microsoft’s older Anomaly Tracking System have just a desktop component.

Once a method to track bugs is implemented you need to evaluate the reported errors and alleged bug reports rapidly. It is important to determine which are real and the priority assigned to the issue. This sets the order the bugs are to be corrected. The reported issues determined to not be real bugs might be a change in requirements or something that needs to be addressed in training. Some reports are questions to be addressed in the Frequently Asked Questions (FAQ) list.

Figure 8. Several bug tracking packages like IssueView have both a desktop and Internet component.

FogBugz (see Figure 9) is one of the leading bug tracking products and is the product I have used at White Light Computing to track bugs my customers send to me since late in 2005. There are several reasons I am using this product:
· SQL Server as the data store

· Accepts email submissions, and informs developers assigned the incident, and the customers who report the incident ongoing status with the incident
· Web interface is clean, responsive, and intuitive
· Highly configurable

· Manages both bugs and enhancement requests

· Keeps track of releases and when bugs are fixed

Figure 9. The FogBugz start up screen.

The user interface is clean and simple and easy to use. It is a rich interface in the Web browser. The hyperlinks take you to different forms and views of the issues (see Figure 10) and the supporting tables. You can configure the settings and the different categories for projects, clients, priorities, and departments. You also can manage email boxes to receive bug report (Figure 11), enhancement requests and inquiries. 
The one feature I really like is the ability to report bugs by email. FogBugz grabs the email from the email server and dumps it into the FogBugz inbox. From the inbox I can manage the email, declare it as spam, or quickly assign it to a developer with a priority, and due dates. 
Another feature I have not seen in other bug tracking software is the concept of snippets. FogBugz snippets are not code, rather they are shortcuts to typing words, phrases, sentences, or even entire paragraphs. The interface is entirely configurable including the snippet, the expanded text, and whether the snippet is global to all users or just a personal snippet for you.

Figure 10. The FogBugz issues list has hyperlinks at the top of the form to access filtering functionality via a menu.

Figure 11. The FogBugz incident form is straightforward and easy to use.

The one big drawback is the lack of reporting. This can be solved with Visual FoxPro and a connection to the FogBugz database. The data model is documented and the connection to SQL Server is easy to establish with Visual FoxPro.

I have evaluated a number of solutions over the years and always found something in the products lacking. This does not mean you will, but it is uncommon to find developers who are completely satisfied with a canned solution. This usually leads developers to create their own solution.
Commercial framework approaches

I find it important to understand how other developers and developer shops approach different aspects of development. It is my hope by understanding different approaches I can refine and better my own code. With this in mind I approached the major commercial framework vendors and asked them to share their wisdom and how they approach error handling and communications of the errors within the frameworks they sell. Each of the following sections contains the responses from the vendors. 
Visual FoxExpress

From Mike Feltman of F1 Technologies (Toledo, Ohio, USA):
There is a global error handler that at runtime resides in the goErrorHandler object. All of the base classes have some default error handling code in their error method - look at cLine.Error or a similar lightweight class for the default implementation. This accomplishes 3 things:

1. Logs all errors to the aErrors array on the object where they occurred. This is primarily handy so that after you step out of the error handling code and back into the offending object you can examine the error.

2. It calls the global error handler, goErrorHandler, which performs logging, asking the developer how to cope with the error (when appropriate) and so on. It makes this call by a macro-substituted call to whatever is defined in ON ERROR.

3. It makes it very easy to augment the error handling in subclasses.

Generally speaking, a subclass will have a series of CASE statements to handle specific errors. If none of the custom handling implemented with the CASE statements does the job, then a DoDefault(...) will defer to the goErrorHandler object. cCursor.Error is a great example of this. 

The Global Error Handler

A global error handler is initialized in the EnableErrorHandling method of the App object. The actual object used for error handling is retrieved from the factory, so it's relatively easy to replace the error handler with one of your own if desired. We have customers that have taken advantage of this to plug in subclasses of our error handler that do things like emailing the developer and that sort of thing. The default class used is cErrorHandler from the cErrors class library. cErrorHandler handles logging, message display (when appropriate) and shut down of the application when necessary. Its Init method takes care of installing a call to it as the global ON ERROR error handler. 

Use of Try/Catch
Several newer classes take advantage of TRY CATCH error handling. This is used primarily when specific methods will possibly throw an error and we want to deal with those errors within the method rather than letting any of the default error handling kick in. (There are just 4 of these that can be found easily with Code References.)

Handling of Expected Errors 

There are lots of errors that are expected to occur as a user works with an application. Basically in VFE these amount to failed database operations due to some issue on the database such as the violation of a candidate or primary key index or a constraint. cCursor has a number of routines designed to deal with errors that may occur when a TABLEUPDATE() or SQLEXEC() fails.

Basically HandleError delegates to specific methods for specific errors. These errors all by pass normal error handling mechanisms and generally just set properties related to the error on the cursor that are retrieved by other objects when it's detected a process (like a TABLEUPDATE() or more generically, as save) failed. 

The other kind of expected errors are when validation fails or a business rule fails. Basically VFE stores error messages of these types in cErrorMessage properties on the offending object(s) and the UI layer makes calls to retrieve and display these messages when it's appropriate. I'm not sure this is within the scope of your session, so I'm not going to go into detail on this for now. Let me know if you want more. 

Miscellaneous

There are other smatterings of error handling throughout the VFE Code. The startup process of a VFE application basically just calls a giant template method that starts services needed by the app. In-line error handling deals with any of these that fail. There are places where errors related to open tables are flagged so other messages that are guaranteed to follow can be suppressed. 

Also there are a number of localized ON ERROR routines where ON ERROR is saved and restored so localized handling could be implemented. Almost all of these contain ON ERROR llError = .T. and I'd guess each of them could be replaced with a TRY CATCH. 

Users can review logged errors using the default Error Log form shown in Figure 12 (or the one enhanced by the developer).

The bulk of the error handling hasn't changed since 1996 or so and I just checked the CSC and there aren't any enhancement requests for changes to it, so it must do a pretty reasonable job. There is a KnowledgeBase article on emailing the ErrorLog table available on the VFE Customer Service Center.

Figure 12. The Visual FoxExpress Error Log form so end users and tech support technicians can review the logged details.

Visual Extend (VFX)
From Uwe Habermann of dFPUG (Kronberg, Germany):
First some information: in several classes there are special error handlers in Error methods, using ON ERROR in older code parts, and using TRY…CATCH in newer code parts of VFX. The error handlers are used when it is expected that the problem can be handled automatically, or is a problem from outside of the application (i.e. ActiveX control cannot be instantiated).

Typical runtime errors are handled in a global error handler. The developer can hook into this mechanism at class level, form level, or redirect the VFX error handler to his own error handler. The standard VFX error handler can be configured. You can set up whether an error dialog similar to the VFP error dialog should appear, or a smarter error dialog similar to other Windows applications, see Figure 13.

Figure 13. This is the enhanced error dialog presented when errors are trapped by a Visual Extend application.
Runtime errors are logged in a table. This may be a DBF, or a SQL Server table. You can configure how detailed the error log should be. The simplest information just contains information about the user (who caused the error), timestamp, VFP error number, and the code which caused the error. The full detailed log level includes all status information like call stack, list memory, list objects and much more. Everything you can get from VFP.

Of course, the developer can configure how the application should behave after an error. (Abort, ignore, retry, user decides)

Runtime errors can be send to the developer by e-mail, or using a web service provided by a server of the developer. The web service project which receives the error reports comes with VFX.

Global error handler
On error is best suitable for a global error handler which catches all errors that are not caught by other special error handlers. I don't see a reason to use on error in other situations since we have try/catch which makes error handling in particular cases much smarter. And this is exactly how it works in VFX except for some older code parts which are not reviewed yet.

TRY/CATCH

TRY…CATCH is a great feature of VFP which allows handling errors within a code block. In my opinion TRY…CATCH should contain as few lines of code as possible and never include any part of the user interface (like MessageBox), or commands like NewObject (the instantiated object should handle errors itself), or procedure/function/method calls (all of them should handle errors themselves).

TRY…CATCH may be used for REPORT FORM, or USE for example. As long as just one line of code is caught, error handling is easy and specific.

Error methods

I just checked some code parts for Error events. Actually, I don't see a reason to use this feature any more. Potential errors should be handled at the point where the problem occurs. Of course, in a strict object oriented environment this event makes it easy to inherit and dedicate error handling. 

So it could be possible to delegate errors to a global error handler in the Error event of the application object. But for functions and procedures you would still need an ON ERROR error handler.
The error handling was improved with each version of VFX. In the last versions me make extensive use of TRY…CATCH to catch any problem at the point where it occurs to avoid bigger trouble. The global error handler was improved to be able to shut down the application properly, even after critical errors, the user dialogs where improved as well as we added features to send error reports by e-mail and web service. All of this was appreciated by developers.

Visual MaxFrame Professional (VMP)
From Art Bergquist of Visionpace (Independence, Missouri, USA):

Visual MaxFrame uses a global error handler installed from the application object:
· oApp::InstallGlobalErrorHandler() calls

· cusInstallErrorHandler::InstallIit() which, in turn, calls

· cusInstallErrorHandler::OnError()

The cusInstallErrorHandler object OnError method initializes the ON ERROR routine:

ON ERROR DO xxerror WITH ERROR(), ;

                         PROGRAM(), ;

                         LINENO(1), ;

                         SYS(16)
All of this is documented in the XXFW.VCX/cusIntsallErrorHandler::zReadMe() method:
Copyright 1997-2005 Visionpace 

All Rights Reserved

    http://www.visionpace.com/vmpsite

    http://vmpdiscussion.visionpace.com

Author:  Drew Speedie

Class:  XXFW.VCX/cusInstallErrorHandler

This custom class definition handles the 

installation of your global ON ERROR routine,

defaulting to the VMP XXERROR.PRG.

It is primarily intended for use in oApp

XXFW.VCX/ctrApp.InstallGlobalErrorHandler(),

but can be used in other situations, too. 

TO IMPLEMENT THIS CLASS

===================================================
1- From an oApp.InstallGlobalErrorHandler()

   LOCAL loInstaller

   loInstaller = CREATEOBJECT("cusInstallErrorHandler",THIS)

   RETURN loInstaller.InstallIt()

2- From a .PRG or other object

   LOCAL loInstaller

   loInstaller = CREATEOBJECT("cusInstallErrorHandler")

   loInstaller.SetErrorLogDir("xxCONFIG.DBF")

   RETURN loInstaller.InstallIt()
Visual MaxFrame Professional does not currently employ the Error event method, but Visual MaxFrame Professional makes extensive use of localized TRY…ENDTRY constructs in the framework and also encourages its localized use in custom/sub-classed code.
With respect to the ErrorLog table itself, here is its structure:

CREATE TABLE (THIS.icErrorLogDir+"ERRORLOG") ;

     (ER_ID I AUTOINC NEXTVALUE 1 STEP 1, ;

      ER_DatTime T, ;

      ER_User    C(5), ;

      ER_UsrName C(40), ;

      ER_Reboot  L, ;

      ER_Error   C(5), ;

      ER_MethPrg C(120), ;

      ER_LineNo  C(5), ;

      ER_Message M, ;

      ER_Code    M, ;

      ER_AError  M, ;

      ER_Program M, ;

      ER_LstMemo M, ;     

      ER_LstStat M, ;

      ER_Config  M, ; 

      ER_Environ M, ;

      ER_Forms   M, ;

      ER_UsInfo  M)

When your application triggers an error you get the System Error screen seen in Figure 14.

Figure 14. The Visual MaxFrame Professional System Error screen allows the user to enter in comments about the problem they encountered.

The error log is euphemistically called the Event log (for users), here’s a short summary of the user option from the first Visual MaxFrame Professional tutorial: Selecting this option instantiates the MYFWFRM.VCX/frmMYEventLog form class.  It allows the Administrator to view the Event Log, which is actually the VMP default ErrorLog.DBF, a free table that is populated with a wealth of error information whenever an error (abnormal event) occurs that is trapped by the default VMP error handler, XXERROR.PRG. There is a sample screenshot of the Event Log from a user perspective in Figure 15.

Figure 15. The Visual MaxFrame Professional Event Log allows users to copy details to the clipboard and later emailed to them to technical support.

Codemine
From Dave Lehr of Soft Classics, Ltd. (Bangor, Maine, USA):
In the CodeMine framework, fatal errors are handled by the Message Manager service object. The Message Manager is a global object used to process all application Information, Warning, Confirmation, and Error messages. The Message Manager takes care of translating message text into the current output language (if applicable), recording messages in the error log, and displaying an appropriate dialog for the message type, when a user interface is available.

The default Fatal Error dialog displays the VFP or User error message, along with a call stack dump, and presents the user with the option to Ignore the error, or Cancel the application.

Object Error Event

All CodeMine foundation classes contain code in their Error event to pass unhandled error events on to the global Message Manager as fatal errors.

Certain objects may convert fatal error signals to a lower severity warning as appropriate. For example, the data object classes treat table open and I/O errors as warnings. 

When an error is handled as a warning, the appropriate message is display in a standard warning dialog by the Message Manager, and the error code is stored in the nLastError property. After each I/O command (USE, SKIP, GOTO, etc) where a handled error may occur, the code will check the status of the nLastError property to determine the success of the command.

ON ERROR handler

A global ON ERRROR handler is also maintained, so that any errors that are signaled outside of any object context are passed directly to the Message Manager as fatal errors.

TRY…CATCH

The TRY…CATCH structured error handling is used for case-specific error handling – that is when you want to apply the special error handling only to the specific code inside the TRY block. This is in contrast to using an object’s Error event, which will get control when a given error occurs anywhere in the object’s code.

An example of a good use for TRY…CATCH would be for evaluating user-defined expressions. Any error (regardless of the error code) in such a user expression would typically display a friendly warning message, and offer the user the chance to correct the expression. If an error occurred in the regular method code of the object, then the normal fatal error processing would still be in effect.

The nErrorMode Property

Each foundation data object has a “nErrorMode” property that gives the developer object level control over how errors are handled in that object. The property can be set to suppress output of messages for warnings, or for all errors that occur in that object.

Errors in Server Applications

Service and COM Server applications will normally have no user interface available. Before displaying any dialogs, the Message Manager will check for the availability of a user interface, using the SYS(2335) function. If no UI is available, no dialog will be displayed.

Use the GetLastMessage method to retrieve the details of the most recent error. For example, if a COM server accessed from an ASP web page encounters an error, it can return the error details for display in the resulting web page output.

Since all management of error and informational messages is done by the Message Manager, the same application object code can be used in interactive applications and server applications with no changes at all. It will be fully interactive when a UI is available and silent when no UI is available.

Error Event Log and Text Log

There are two types of error and message logging supported by the default Message Manager.

The system event log is a table (DBF or SQL Server) based log, where all fatal errors and related stack dump information are stored. This log can be enabled or disabled by user preference.

The text log feature is intended mostly for service applications, where no UI is available. Although, the text log feature may be enabled for interactive applications as well.

When the text log is enabled, all messages (warning, confirmation, info, fatal, etc) are written to the text log. The text log supports normal and verbose modes. Developers can use the text log in normal mode to keep track of how their service is performing. Verbose mode can be used to output additional information that would be useful in diagnosing problems.

Mere Mortals for VFP
From Paul Mrozowski (MM - MVP) for Oakleaf Enterprises (Charlottesville, Virginia, USA):
Mere Mortals (MM) implements error handling at the global and at the class level. The primary error handling class, CErrorMgr (CManager.VCX) is instanciated by (and attached to) the application object when it starts up. MM uses a table-based abstract factory to determine the class that gets instantiated. This makes it relatively easy to replace the error handler with any custom handler you’d like, as long as you implement the same interface.

An object reference to the error handler is accessible via goApp.oErrorMgr anywhere within the application. In the Init() of this class, the existing ON ERROR handler is saved and ON ERROR then gets pointed to the ErrorHandler method of this object (passing in the error number, program, and line number). The default error handler is restored when this object is destroyed.

When an error occurs, the ErrorHandler method first calls the HandleError method, which gets a chance to deal with the error (or, in some cases, ignore the error because it should be handled by another object - this might be the case for things like database conflicts). If this method returns true, the error is assumed to be handled and we're returned out of the ErrorHandler method. Otherwise, it grabs the error number, program, line number, error message, executing program name and logs the error along with various environment settings (OS, processor, disk space, machine name, user ID, etc.). This information gets logged into a VFP table. The error message is then displayed. If you are running in development mode, it will give you the opportunity to activate the debug window.

In the base form class, the framework has code in the Error method which checks to see if the goApp.oErrorMgr object is available. If so, it passes the error off to it. Otherwise it just raises the standard VFP error handler.

In the base container class, the Error method walks up the container hierarchy to determine if one of those control's can handle the error. If so, the error is passed up (all the way up to the form level). Once it reaches the form level, if the global error handler object is found in goApp.oErrorMgr, it is called. Otherwise the default VFP error handler is called. 

The base business object class has a bit more code to handle errors. It's important to note that the base business objects are dropped on the base container class, so if a business object is dropped on a form it's possible to handle errors that were raised by the object in the form itself. The code here determines if the error originated inside the business object. If so, it instantiates the error handler class specified in it's cErrorClass property, then calls it's HandleError method. It also determines if it's being run as a COM object, in which case it grabs some basic error information, logs it to a text file, and also adds it to the object's aErrorMsg array property. Otherwise, it passes the error up the inheritance chain. Again, since the business object is contained in a container object, the error progression follows the same sequence as the base container class from this point.

There are also Cursor, DataEnvironment, and behavior objects (among others) which all have their own Error handler code. Each follows the same error handling pattern: You get the opportunity to deal with the error as close to the source as possible. If you can't, the error gets passed further up the chain until the global handler kicks in and logs/reports the error.

The framework does not use or take advantage of the new TRY/CATCH error handling available in VFP 8 and 9.

There is a built-in form which lets you (or your users) review any errors that have been logged (see Figure 16).

Figure 16. The Mere Mortals Error Log allows users to print a report, and clean up the table when errors are reported.

Promatrix
From Joe Lawson of Promatrix Corporation (Tampa, Florida, USA)
Go here and download the VPME documentation and read all the sections about the VPME errorhandler: http://www.promatrix.com/vpmdocs/docsdown.htm 

Also, download the VPME Demo and the Sample Application to see the error handler interfaces in a VPME application: http://www.promatrix.com/demos/demos.htm 

If you have any specific questions, post them on the ProMatrix Developers Newsgroup: http://www.promatrix.com/news/newsgroups.htm 
Resources

There are a lot of resources out on the Web and in books, but here are a few I find useful:

1. The VFP Help file has several excellent topics
2. Error Handling in Visual FoxPro, by Doug Hennig, available online at http://stonefield.com
3. Error Handling in VFP 8, by Doug Hennig from conference CD/downloads
4. Structured Error Handling in VFP 8, by Markus Egger, CoDe FOCUS: Visual Foxpro 8 (March 2003), and available online http://code-magazine.com
5. Exception Handling Dilemma in VFP, Rick Strahl’s WebLog (http://west-wind.com/weblog) February 9, 2006 post
6. Debugging Visual FoxPro Applications, by Nancy Folsom, Hentzenwerke Publishing
7. FoxPro Wiki, many topics and authors, available online http://fox.wikis.com 

Special thanks

This session was assembled with a lot of help from several generous people in the Fox Community. When Whil asked me to take on best practices on error handling and reporting I thought to myself: I have not really dealt about error handling in years because framework vendors do this thinking for me. It forced me outside of the box of my normal session and made me revisit the topic I have sort of forgotten about.

Doug Hennig presented many sessions on error handling, and has a whitepaper on Stonefield.com everyone points to when developers are looking for help with error handling on the various community forums. Doug’s paper was the first thing I downloaded and read before preparing for this session. You might consider doing the same if error handling is of interest to you at some point in the future.
One of the great things about writing a conference session or an article is digging into the topic so you can discuss things accurately and efficiently. When you consider writing a topic on best practices on any subject you revisit the way you address the subject matter. In the case of error handling I decided early on I needed to understand how framework vendors address error handling in their frameworks. I only have to address error handling to my framework and the way I design and implement my code. Framework vendors have to address error handling to adapt to the way all their customers code. This is a big challenge. With this in mind I asked the framework vendors to help me out, and in effect help me write part of this whitepaper. Their generosity is much appreciated and their feedback was a big help.

Rick Borup is someone I am lucky enough to call a friend. His generosity does not have any bounds. His contributions to the Fox Community have helped me professionally. In particular he shared his knowledge of RSS and this really inspired me to finally push and accomplish adding RSS feeds in my error handler to ease the communications when it comes to errors in my applications.

The Grand Rapids Area Fox User Group (GRAFUG) is one of my favorite groups to present to. The members provide excellent feedback and have often been the guinea pigs for my rehearsals. They were the only group I presented this session to before it was given to a conference audience and it helped me refine the session.
Conclusion/Summary

Error handling is a necessary evil in software development. I hope this session and whitepaper shed some light on best practice approaches, architecture, logging, reporting issues, and tracking errors created in your applications.
Unfortunately there is no way in a short whitepaper or a 90 minute conference session to cover and address every single error handling condition and remedy. You will have to look into your application and apply the best practices as needed. This whitepaper does not cover all possible error handling conditions and your application might have special considerations based on the industry it serves and the customers you developed the application for. 
If you recognize something I have overlooked, please send it to me so I can improve the session and the whitepaper for future presentations. My email and contact information are on the first page of this whitepaper.

